

SAS Technical News

Spring 2009

*For Higher
Customer Satisfaction,
We Bridge
the SAS System
Between
Customer's World.*

CONTENTS

- 1 SAS 9.2への移行と活用
- 6 SAS Academic News
- 11 Q&A
- 14 SASトレーニングのお知らせ
- 16 最新リリース情報
- 16 SAS Technical News送付についてのご案内

特集

SAS 9.2への 移行と活用



1. はじめに

2005年、SASプロダクトの正常進化系としてBusiness Intelligenceを実現するSAS BIサーバーをリリースしました。リリース当初のSAS BIサーバーは、SAS 9.1を基本モジュールとして構成されておりました。2008年9月、SAS全体を支える基本モジュールであるSAS 9.2が新機能を搭載しリリースされたことにより、SAS BIサーバーは「見える化」から「予見力」へと更なる発展を遂げ、SASは世界のBIのトップベンダーとして進化し続けています。

SAS Technical Newsでは、SAS 9.2の機能に特化した「プラットフォーム編」、「Foundation編」、「SAS 9.2への移行について」をご紹介してまいりました。本号では、これらに続く最終フェーズとして、SAS 9.2以前のバージョンで作成されたSASファイルをSAS 9.2で活用する方法をご紹介いたします。なお、SAS 9.2のご紹介として掲載させていただきました過去3部作の特集記事は、弊社ホームページの以下のリンクから入手可能ですので、合わせてご参照ください。

Summer 2008 SAS 9.2のご紹介（プラットフォーム編）

特集の概要：SAS 9.2のBIサーバーについてのプロダクトコンセプト、BIサーバーシステム構成図及びユーザから見た業務別SAS BIサーバー活用例（ユーザーマッピング）のご紹介
<http://www.sas.com/japan/periodicals/technews/pdf/08sum.pdf>

Autumn 2008 SAS 9.2のご紹介（Foundation編）

特集の概要：SAS 9.2基本機能（Foundation）に特化した新機能、拡張プラットフォーム種類、また特集2として、システム必要条件に関するSAS 9.2インストールセンターのご紹介
<http://www.sas.com/japan/periodicals/technews/pdf/08aut.pdf>

Winter 2009 SAS 9.2への移行について(事前準備編)

特集の概要: 移行計画を立案する上での事前準備、環境の構築、
実施作業の概要についてご紹介

<http://www.sas.com/japan/periodicals/technews/pdf/09win.pdf>

2. 既存資産の活用方法および確認事項

まず、旧SASバージョンであるSAS 6、SAS 8、SAS 9.1.3で作成したSASファイルをSAS 9.2で活用するには、旧SASバージョン形式のままSAS 9.2で使用するか、または全てSAS 9.2形式へ移行するかを選択することができます。この章では、この2つの活用方法についてご説明いたします。また、実作業を行うにあたっての確認必要事項についてご説明します。

2.1 既存資産の活用方法

SAS 9.2での既存資産の活用は、以下の2つの方法があります。

- ・SASファイルを旧バージョン形式のまま、SAS 9.2にて活用する
- ・SASファイルを全てSAS 9.2形式へ移行する

前者の旧バージョン形式のままSAS 9.2で活用する場合、基本的にSAS 9.2への移行作業を実施することなくSASファイルを参照できますので、移行に関わる作業を最小限にとどめることができます。ただし、SASファイルの種類によっては、サポートされる処理が制限されます。

次に、後者である旧バージョン形式のSASファイルをSAS 9.2形式へ移行する場合、移行用のプロシジャにて作業を行います。この場合、移行対象のファイルの選別から移行に関わる準備、処理結果の確認作業が発生するため、ある程度の時間と作業に関わる移行計画の立案が必要となるのが想定されます。

SASファイルをSAS 9.2形式に移行することと前者が大きく異なる点は、先に述べたSASファイルへの書き込みや更新処理に使用制限が発生しないことです。一般的に、通常のSASアプリケーションでは、SASファイルの読み込み、更新、新規作成を行なっているかと思えます。旧バージョン形式のままSASファイルを活用した場合、処理に制限が発生し、結果としてSASアプリケーションの動作に支障を来す恐れがあります。このため移行に関するコストや時間が必要になりますが、現実的にSASアプリケーションの処理内容を想定するとSAS 9.2形式へ移行することを強く推奨いたします。

2.2 移行対象SASファイルの確認

実作業を行う前に、既存資産のSASファイルの種類を確認します。SASファイルとは、SAS独自の形式で保存されたファイル群で、主なものには以下のものがあります。

| ファイルの種類 | 概要 |
|---------------|--|
| データセット | オブザベーション群および管理情報 |
| DATAステップビュー | DATAステップにてVIEW=オプションを使用した際に作成され、DATAステップのプログラムが格納されたファイル |
| SAS/ACCESSビュー | ACCESSプロシジャにて作成され、データベースのテーブルを読み込む際に使用する |
| SQLビュー | SQLプロシジャにて作成され、SQLのクエリが格納されたファイル |
| SASカタログ | 主にFORMATプロシジャにて作成され、入力・出力形式が格納されたファイル |

SASファイルの種類は、そのファイルの拡張子から確認できます。SASのバージョンおよび稼働プラットフォームにより異なりますので、次の表をご確認ください。

Windows版SASの拡張子対応表

| ファイルの種類 | SAS 6での拡張子 | SAS 8、9での拡張子 ¹ |
|-----------------------|------------|---------------------------|
| データセット | .sd2 | .sas7bdat |
| DATAステップビュー SQLビュー | .sv2 | .sas7bvew |
| ACCESSビュー | .sa2 | .sas7bacs |
| SASカタログ | .sc2 | .sas7bcac |

1: SAS 8以降はすべてのプラットフォームにて共通の拡張子になります。

UNIX版の拡張子対応表

| ファイルの種類 | SAS 6での拡張子 ² | SAS 8、9での拡張子 ¹ |
|-----------------------|-------------------------|---------------------------|
| データセット | .ssd0nn | .sas7bdat |
| DATAステップビュー SQLビュー | .ssv0nn | .sas7bvew |
| ACCESSビュー | .ssa0nn | .sas7bacs |
| SASカタログ | .sct0nn | .sas7bcac |

1: SAS 8以降はすべてのプラットフォームにて共通の拡張子になります。

2: nnはオペレーティングシステム(以下、OS)毎の固有の番号です。

| nn | 対応 OS |
|----|-------------------|
| 01 | Solaris、HP-UX、AIX |
| 04 | Tru64 UNIX |

2.3 エンジンについて

SASはエンジンを通じて、SASファイルの読み込み、書き込み、更新処理を行ないます。SASファイルが作成されたバージョンと、それを利用するSASのバージョンが異なる場合には、エンジンを指定しファイルにアクセスします。

エンジンの指定例:

LIBNAME libref V6 'SAS6にて作成されたSASファイルの格納ディレクトリ名';

LIBNAME libref V8 'SAS8にて作成されたSASファイルの格納ディレクトリ名';

LIBNAME libref V9 'SAS9にて作成されたSASファイルの格納ディレクトリ名';

2.4 Native、Non-Nativeについて

旧バージョンで作成されたSASファイルをSAS 9.2にて活用または移行する際、対象のSASファイルとSAS 9.2の互換性を識別するため、Native、Non-Nativeという言葉でカテゴリ化する必要があります。つまり、Native、Non-Nativeの定義とは、SAS 9.2が稼働するプラットフォーム等のシステム環境を基準として、対象のSASファイルが同一のシステム環境で作成されたものであるかを示します。

Nativeの条件

- ・OSファミリーが同一である
- ・SASのbitが同一である
- ・SASのエンコードが同一である

これらの条件がひとつでも異なる場合、Non-Nativeとなります。

Non-Nativeの例

- ・Windows版SAS 6で作成したデータセットを、x64 Windows版SAS 9.2で活用する
- ・Solaris版SAS 6(32bit)で作成したデータセットを、Solaris版SAS 9.2(64bit)で活用する
- ・Solaris版SAS 6 SJISで作成したデータセットを、Solaris版SAS 9.2 EUCで活用する

特にSASファイルをSAS 9.2形式へ移行する場合、対象のSASファイルがどのバージョンのSASで作成されたものなのか、さらに移行先環境のSAS 9.2から見た場合、それらのSASファイルがNativeなのかNon-Nativeであるかによって、移行作業にて使用するプロシジャが異なります。次のフローチャートを参照し、移行対象となるSASファイルがNativeかNon-Nativeなのかを分類してください。




図 1: Native、Non-native分類のフローチャート

3. 既存資産の活用

旧バージョン形式のSASファイルをSAS 9.2形式に移行せずに活用する場合、そのファイルが作成されたSASのバージョン、処理内容およびSAS 9.2から見てNativeまたはNon-Nativeの分類によって、サポートされる処理が制限される場合があります。以下では、その詳細をご紹介します。

3.1 SASファイルへの処理内容の分類

SASファイルへの処理内容は、3つに分類されます。以下の表は、データセットへの処理を例としています。

| 記号 | 処理内容 | 具体例 |
|----|------|--|
| R | 読み取り | PRINT プロシジャにてデータセットの内容を表示する DATA ステップのSETステートメントにてデータセットを読み込む |
| U | 更新 | APPEND プロシジャにてデータセットにデータを追加する |
| W | 書き込み | データセットを新規に作成する |

3.2 SAS 9.2 での旧バージョン形式のSASファイルの取り扱い

旧バージョン形式のSASファイルを、SAS 9.2にて活用する場合にサポートされる処理内容は以下のとおりです。

SAS 6形式ファイルのSAS 9.2での扱い (R:読み取り U:更新 W:書き込み)

| ファイルタイプ | Native | | | Non-Native | | |
|---------------|--------|---|---|------------|---|---|
| | R | U | W | R | U | W |
| データセット | | | | 1 | x | x |
| DATAステップビュー | | x | x | x | x | x |
| SAS/ACCESSビュー | 2 | 2 | x | x | x | x |
| SQLビュー | | x | x | x | x | x |
| SASカタログ | | x | x | x | x | x |

1:同一OSファミリー及びエンコードで作成されたファイルのみ対象です。なお、Solaris x64版SAS 9.2では、SAS 6形式のファイルを活用することは出来ません。
2:ACCESSビューは、SAS 8以降の新機能を利用することができません。

SAS 8形式ファイルのSAS 9.2での扱い (R:読み取り U:更新 W:書き込み)

| ファイルタイプ | Native | | | Non-Native | | |
|---------------|--------|---|---|------------|---|---|
| | R | U | W | R | U | W |
| データセット | | | | 3 | x | 3 |
| DATAステップビュー | | x | x | x | x | x |
| SAS/ACCESSビュー | 4 | 4 | x | 5 | x | x |
| SQLビュー | | | x | | x | x |
| SASカタログ | | | | x | x | x |

3:同一エンコードで作成されたファイルが対象です。また、64bit OSにて稼働するSAS 9.2にて新規作成した場合、64bitビットのファイルが作成されます。

4:SAS/ACCESSビューはSAS 8以降の新機能を使用することができません。

5:SAS/ACCESS OracleインターフェイスおよびSybaseインターフェイスによって作成されたファイルのみがサポート対象です。

なお、SAS 9.1.3にて作成したSASファイルをSAS 9.2にて活用する場合、サポートされる処理内容はSAS 8のものと同様です。

4. 既存資産の移行

SAS 9.2の機能をすべて利用するためには、旧バージョン形式のSASファイルをSAS 9.2形式に移行する必要があります。移行方法は、移行対象のSASファイルの種類および同ファイルがSAS 9.2から見てNativeまたはNon-Nativeに分類されるのかによって異なります。以下のフローチャートを使用して、適切な移行方法を確認ください。




図 2: SAS 9.2形式にSASファイルを移行する方法

移行方法の詳細に関しては、以下をご確認ください。

MIGRATE プロシジャ : 4.1および4.2

MIGRATE プロシジャ(リモートオプションあり) 3) : 4.1および4.4

CPORT/CIMPORT プロシジャ : 4.3

3 リモートオプションの詳細は、後述します。

4.1 MIGRATE プロシジャの概要および注意事項

MIGRATE プロシジャは、既存のSASファイルを最新の形式に変換するためのプロシジャです。MIGRATE プロシジャの構文は以下の通りです。

```
LIBNAME <移行元ライブラリ名> <エンジン名>
'移行元ライブラリのパス';
LIBNAME <移行先ライブラリ名> v9
'移行先ライブラリのパス';
PROC MIGRATE IN=source OUT=target <オプション>;
RUN;
```

移行先ライブラリのパスには、必ず新規に作成した空のディレクトリを指定します。これは、MIGRATE プロシジャは、既にSASファイルが存在した場合、以下のエラーが発生し不適切なファイルの上書きを防ぐよう設計されています。

```
ERROR: File TARGET.XXXXX (memtype=ZZZZ) cannot be migrated.
A file of that name already exists in the OUT= library TARGET.
```

なお、Non-Nativeの移行する場合、以下のNOTEおよびWARNINGメッセージが出力されますが、処理自体に問題はありません。

```
NOTE: データファイル XXX.YYY.DATA は別なホストにネイティブな形式が使用されているか、またはエンコーディングがセッションエンコーディングと一致していません。クロス環境データアクセスが使用されるため、追加のCPUリソースが必要となり、パフォーマンスが低下します。
```

```
WARNING: クライアントセッションのエンコーディングは xxxxxx ですが、サーバーセッションには未サポートのエンコーディングが使用されています。データは正しく転送されません。
```

その他、MIGRATE プロシジャでは、以下のSASファイルを移行することが出来ません。

- ・AIX版SAS 6のSASカタログ
- ・Tru64 UNIX版のSASカタログのx64 Linux版またはx64 Solaris版への移行
- ・Linux(Itanium)版のSASカタログのx64 Linux版またはx64 Solaris版への移行
- ・ソースのないIDATAステップビュー
- ・SAS/ACCESS OracleインターフェイスおよびSAS/ACCESS Sybaseインターフェイス以外で作成されたACCESSビュー
- ・マクロカタログ
- ・PROFILEカタログ
- ・SAS 6.12以前のリリースにて作成されたSASファイル

上記のAIX版SAS 6のSASカタログ、Tru64 UNIX版のSASカタログおよびLinux(Itanium)版SAS 9.1.3のSASカタログは、CPORT/CIMPORT プロシジャを使用して移行してください。その他のファイルに関しては、SAS 9.2にて再作成する必要があります。

4.2. MIGRATE プロシジャによる移行方法

NativeファイルおよびNon-Nativeファイルのデータセットのみを移行する場合は、MIGRATE プロシジャを使用します。以下の環境を例に、詳細をご紹介します。

移行元、移行先ディレクトリ一覧

```
移行元
ディレクトリ      : c:\%saslib%\source
ライブラリ参照名 : source

移行先
ディレクトリ      : c:\%saslib%\target
ライブラリ参照名 : target
```




図3: MIGRATE プロシジャによる移行方法

もし、移行元と移行先が異なる筐体の場合は、移行先にファイルを転送します。なお、FTPを使用してファイルを転送する場合は、必ずバイナリモードを使用してください。具体的な手順は以下のとおりです。

- 1) SAS 9.2を起動し、移行元のライブラリを定義します。この際、SAS 8にて作成されたSASファイルを移行する場合は、V8エンジンを指定します。SAS 6にて作成されたSASファイルを移行する場合は、V6エンジンを指定します。

```
LIBNAME source <v6/v8> 'c:\%saslib%\source';
```

- 2) 空のディレクトリを作成し、V9エンジンのSASライブラリを定義します。

```
LIBNAME target v9 'c:\%saslib%\v92lib';
```

- 3) MIGRATE プロシジャを実行します。

```
PROC MIGRATE IN=source OUT=target;
RUN;
```

4.3 CPORT/CIMPORT プロシジャ

Non-NativeのデータセットおよびSASカタログのみを移行する場合は、CPORT/CIMPORT プロシジャを使用します。次の環境を例に、詳細をご紹介します。

移行元、移行先ディレクトリ一覧

| | |
|----------|-------------------|
| 移行元 | |
| ディレクトリ | : /migrate/source |
| ライブラリ参照名 | : source |
| 移送ファイル名 | : tran.cpt |
| 移行先 | |
| ディレクトリ | : /migrate/target |
| ライブラリ参照名 | : target |

具体的な手順は以下のとおりです。

- 1) 移行元のSAS 6またはSAS 8を起動し、CPORTプロシジャを使用して移送ファイルを作成します。

```
FILENAME tran '/migrate/tran.cpt'; /* 移送ファイル*/
LIBNAME source '/migrate/source';
PROC CPORT LIB=source FILE=tran;
RUN;
```

- 2) 移行元と移行先が異なる筐体の場合は、移行先にファイルを転送します。なお、FTPを使用してファイルを転送する場合は、必ずバイナリモードを使用します。

- 3) 移行先のSAS 9.2を起動し、空のディレクトリを作成して、V9エンジンのSASライブラリを定義します。その後、CIMPORTプロシジャにて移送ファイルを復元します。

```
LIBNAME target v9 '/migrate/target'; /*v9エンジンのライブラリ*/
FILENAME tran '/migrate/tran.cpt'; /* 移送ファイル*/
PROC CIMPORT LIB=target INFILE=tran;
RUN;
```

4.4 MIGRATEプロシジャ(リモートオプションあり)による移行方法

Non-NativeのSASファイルには、SAS 9.2より直接読み取りが出来ないものがあります。そのため、Non-NativeのSASファイルを可能な限り移行するには、SAS 9.2からSAS 8にSAS/CONNECTの機能を利用して接続し、SAS 9.2から透過的にSAS 8上のライブラリが参照可能となるようリモートライブラリサービス(以下、RLS)を使用します。なお、同ライブラリ参照名は、サーバー上のライブラリ参照名を指定するためのMIGRATEプロシジャのSLIBREFオプションに指定します。以下の環境を例に、詳細をご紹介します。

| | |
|--------------|-------------------|
| 移行元 | |
| RLSのディレクトリ | : /migrate/source |
| RLSのライブラリ参照名 | : rlslib |
| 移行先 | |
| ディレクトリ | : /migrate/target |
| ライブラリ参照名 | : target |




図 4: MIGRATE プロシジャ(リモートオプションあり)による移行方法

具体的な手順は以下のとおりです。

- 1) SAS 9.2より、SAS/CONNECTの機能を利用してSAS 8に接続します。

```
OPTIONS REMOTE=<リモートホスト名> COMAMID=TCP;
FILENAME RLINK 'スクリプトファイルのパス';
SIGNON;
```

- 2) SAS 8のリモートセッションにて、移行元のライブラリを定義します。この際、SAS 8にて作成されたSASファイルを移行する場合は、V8エンジンを指定します。SAS 6にて作成されたSASファイルを移行する場合は、V6エンジンを指定します。

```
RSUBMIT;
LIBNAME rlslib <v6/v8> '/migrate/source';
ENDRSUBMIT;
```

- 3) SAS 9.2のセッションにて、上記2で定義した同一ディレクトリに異なるライブラリ参照名を定義します。この際、SAS 8にて作成されたSASファイルを移行する場合は、V8エンジンを指定します。SAS 6にて作成されたSASファイルを移行する場合は、V6エンジンを指定します。なお、移行元と移行先環境が別筐体の場合は、移行元のSASファイルが移行先環境から参照できるようにNFS等を利用してファイル共有し、ライブラリを定義します。

```
LIBNAME source <v6/v8> '/migrate/source';
```

- 4) SAS 9.2のセッションにて、RLSを使用したライブラリを定義します。

```
LIBNAME rlslib SERVER=sassrv;
```

- 5) 空のディレクトリを作成し、SAS 9.2のセッションにてV9エンジンのライブラリを定義します。

```
LIBNAME target v9 '/migrate/target';
```

- 6) SLIBREFオプションにRLSを使用したライブラリ参照名を記述し、MIGRATEプロシジャを実行します。

```
PROC MIGRATE IN=source OUT=target SLIBREF=rlslib;
RUN;
```

5. SAS 9.2への移行時の注意点

SAS 9.2のSAS/ACCESS Oracleインターフェイスにて、ORACLEのデータ型がchar, varchar, varchar2, nchar, nvarcharのデータを読み込むと、SASの変数長が旧バージョンのものと比較して2倍になります。ADJUST_BYTE_SEMANTIC_COLUMN_LENGTHS=NO オプションを、LIBNAMEステートメントまたはSQLプロシジャのCONNECTステートメントに追加することで回避いただけます。本事象の詳細に関しては、下記URLをご参照ください。

Problem Note 32590:

Data might be truncated in columns CHAR & VARCHAR2 in UTF-8 SASR 9.1 sessions

<http://support.sas.com/kb/34/590.html>

Hotfix E9BB50が適用されていないSAS 9.1.3にてCPORTプロシジャにて移送ファイルを作成後、SAS 9.2にてCIMPORTプロシジャを実施すると、以下のエラーメッセージが表示されます。SAS 9.1.3にてHotfix E9BB50を適用後、CPORTプロシジャにて移送ファイルを再作成ください。本事象の詳細に関しては、下記URLをご参照ください。

Problem Note 20578:

PROC CIMPORT generates warning: "This transport file is from an earlier SASR release"

<http://support.sas.com/kb/20/578.html>

SAS 9.2への移行についての技術的なお問い合わせいただく際は、以下の情報を添えてFAXまたはE-mailにて弊社テクニカルサポートにお問い合わせください。

- ・移行元のOS名、SASのバージョン
- ・移行先のOS名、SASのバージョン
- ・実行ログ

お問い合わせ方法の詳細に関しては、下記URLをご参照ください。


<http://www.sas.com/japan/service/technical/techsup.html>

6. おわりに

今回の特集の焦点といたしましては、SAS 9.2から旧SASバージョンで作成されたSASファイルをどのように扱うのか、また旧SASバージョンのSASファイルをどのようにSAS 9.2へ移行するかを具体的に説明しております。

多くのお客様に最新版のSASシステムをご利用いただくため、改善すべき点はまだまだありますが、移行ツールのご提供によるお客様の作業負担の軽減を実現できることを念頭に進化しております。過去の移行に関する特集を踏まえ、SAS 9.2への移行についてご参考いただけますようお願いいたします。

SAS Academic News



はじめに

2009年春号から、福岡大学商学部 杉本准教授によるSAS Enterprise Guideを活用した事例紹介を4部作で連載いたします。その他、SASグローバル認定プログラムのサンプル問題の一例、最後に東京医科歯科大学の大橋助教によるSAS四方山話「アンケートバイアスと基本集計の方法」をお届けいたします。

事例紹介

SASグローバル認定プログラムサンプル問題

コラム「SAS四方山話」

事例紹介



杉本 宏幸
福岡大学商学部准教授
博士(商学)



福岡大学は10学部約2万人の学生を擁する大きな組織で、本年は創立75周年記念の節目の年にあたります。私が所属する商学部商学科は、マーケティング・流通のスタッフが私を含め6名と多く、研究・教育環境として恵まれてはいますが、様々な教育面での課題に直面しています。

その一つに、ビジネスの現場と教育をいかに連携できるかという問題があります。特に九州・福岡では通販会社をはじめ、ビジネスの現場でデータ分析のニーズが強く、地場企業に対してもデータ分析ができる人材の育成・提供が急務の課題として浮上しています。しかし、正直にいえば、本学部ではこの分野はまだメジャーでなく、データ分析という武器を身につけて卒業する学生は必ずしも多くないというのが、現実です。

現在、マーケティングを専攻する学生達に定量的な考え方を養って

もらおうと、商学科のスタッフ数名で取り組みをはじめました。調査やデータ分析を一定程度、経験・習得する学生が学科の約10%(一学年の定員210名)となることを目標に、2名のスタッフが講義・ゼミを中心にデータ分析に親しんでもらう試みを行っています。SAS、SAS Enterprise Guideは私の3年生ゼミ(18名)で使用しており、本年9月からの講義でも使用予定です。

商学部では、分析結果をどう解釈・活用するかという点も重視すべきと考えます。これには他科目で得られる知識等が重要なため、多様な科目の連携が不可欠と思われます。こうした動きが学科・学部全体に広がれば、多くの学生達が実践に必要なデータ分析の能力を身につけてくれるのではと構想しています。

福岡大学商学部の試みはまだ着手しはじめたばかりです。我々も学生達も手探り状態というのが本音です。しかし、今回せっかくの機会を頂戴しましたので、本学商学部の試みと苦闘する我々の姿に、一年間お付き合い下さればと思っております。

SASグローバル認定プログラムサンプル問題

近年、SAS認定試験プログラムを取得するお客様が増えてきており、ワールドワイドで2万人の方々SAS認定プログラムを取得しております。今回は、SAS認定プログラムのいくつかのコースの中から、以下のベーシックコース及びアドバンスコースをピックアップし、実際にどのような内容の問題であるのかをご紹介します。なお、問題の解答は、章の最後に記載しております。

SAS Base Programming for SAS 9

SAS Advanced Programming for SAS 9

SAS Base Programming for SAS 9

Item 1

The following program is submitted.

```
data test;
  input name $ age;
cards;
John +35
;
```

run;

Which values are stored in the output data set?

- A. name age

John 35
- B. name age

John (missing value)
- C. name age

(missing value) (missing value)
- D. The DATA step fails execution due to data errors.

Item 2

The following observation is stored in a SAS data set named EMPLOYEES:

```
LNAME FNAME JOBCODE
-----
Whitley Sam na1
```

If the DATA step below is executed, what will be the value of the variable JOBDESC in the output SAS data set when this observation is processed:

```
data navigate;
  set employees;
  if jobcode = 'NA1' then jobdesc = 'Navigator';
run;
```

- A. navigator
B. Navigator
C. NAVIGATOR
D. a missing value

Item 3

The following SAS program is submitted:

```
proc format;
  value score 1 - 50 = 'Fail'
             51 - 100 = 'Pass';
run;
```

Which one of the following PRINT procedure steps correctly applies the format?

- A. proc print data = sasuser.class;
var test;
format test score;
run;
- B. proc print data = sasuser.class;
var test;
format test score.;;
run;
- C. proc print data = sasuser.class format = score;
var test;
run;
- D. proc print data = sasuser.class format = score.;;
var test;
run;

Item 4

Given the following DATA step:

```
data loop;
  x = 0;
  do index = 1 to 5 by 2;
  x = index ;
  end;
run;
```

Upon completion of execution, what are the values of the variables X and INDEX in the SAS data set named LOOP?

- A. x = 3, index = 3
B. x = 3, index = 4
C. x = 5, index = 5
D. x = 5, index = 6
E. x = 5, index = 7

Item 5

Given that the data set named ONE contains 10 observations and the data set named TWO contains 10 observations, how many observations will be contained in the data set named COMBINE that is created in the following DATA step?

```
data combine;
  set one two;
run;
```

- A. 10
- B. 20
- C. 0, the DATA step will fail due to syntax errors
- D. 10 to 20, depending on how many observations match

SAS Advanced Programming for SAS 9

Item 1
Given the following SAS data sets ONE and TWO:

| ONE | | | TWO | | |
|------|-----|--------|------|-----|-------|
| YEAR | QTR | BUDGET | YEAR | QTR | SALES |
| 2001 | 3 | 500 | 2001 | 4 | 300 |
| 2001 | 4 | 400 | 2002 | 1 | 600 |
| 2002 | 1 | 700 | | | |

```
The following SAS program is submitted:
proc sql;
  select one.*, sales
    from one right join two
      on one.year = two.year;
quit;
```

Which one of the following reports is generated?

- A.

| YEAR | QTR | BUDGET | SALES |
|------|-----|--------|-------|
| 2001 | 3 | 500 | . |
- B.

| YEAR | QTR | BUDGET | SALES |
|------|-----|--------|-------|
| 2001 | 4 | 400 | 300 |
| 2002 | 1 | 700 | 600 |
- C.

| YEAR | QTR | BUDGET | SALES |
|------|-----|--------|-------|
| 2001 | 3 | 500 | . |
| 2001 | 4 | 400 | 300 |
| 2002 | 1 | 700 | 600 |
- D.

| YEAR | QTR | BUDGET | SALES |
|------|-----|--------|-------|
| 2001 | 3 | 500 | 300 |
| 2001 | 4 | 400 | 300 |
| 2002 | 1 | 700 | 600 |

Item 2
Given the following SAS data sets ONE and TWO:

| ONE | | TWO | |
|-----|-------|-----|-------|
| NUM | CHAR1 | NUM | CHAR2 |
| 1 | A | 2 | X |
| 2 | B | 3 | Y |
| 4 | D | 5 | V |

```
The following SAS program is submitted creating the
output table THREE:
data three;
merge one (in = in1) two (in = in2);
  by num;
run;
```

| THREE | | |
|-------|-------|-------|
| YEAR | CHAR1 | CHAR2 |
| 1 | A | |
| 2 | B | X |
| 3 | | Y |
| 4 | D | |
| 5 | | V |

Which one of the following SQL programs creates an equivalent SAS data set THREE?

- A.

```
proc sql;
  create table three as
  select *
    from one full join two
      where one.num = two.num;
quit;
```
- B.

```
proc sql;
  create table three as
  select coalesce(one.num, two.num)
    as NUM, char1, char2
    from one full join two
      where one.num = two.num;
quit;
```
- C.

```
proc sql;
  create table three as
  select one.num, char1, char2
    from one full join two
      on one.num = two.num;
quit;
```
- D.

```
proc sql;
  create table three as
  select coalesce(one.num, two.num)
    as NUM, char1, char2
    from one full join two
      on one.num = two.num;
quit;
```

Item 3
The following SAS program is submitted:

```
%let type = RANCH;
proc sql;
  create view houses as
  select *
    from sasuser.houses
  where style = "&type";
quit;
```

```
%let type = CONDO;
```

```
proc print data = houses;
run;
```

The report that is produced displays observations whose value of STYLE are all equal to RANCH.

Which one of the following functions on the WHERE clause resolves the current value of the macro variable TYPE?

- A. GET
- B. SYMGET
- C. %SYMGET
- D. &RETRIEVE

Item 4
The SAS data set SASDATA.SALES has a simple index on the variable DATE and a variable named REVENUE with no index.

In which one of the following SAS programs is the DATE index considered for use?

- A. `proc print data = sasdata.sales;
by date;
run;`
- B. `proc print data = sasdata.sales;
where month(date) = 3;
run;`
- C. `data march;
set sasdata.sales;
if '01mar2002'd < date <
'31mar2002'd;
run;`
- D. `data march;
set sasdata.sales;
where date < '31mar2002'd or revenue
> 50000;
run;`

[correct_answer]

SAS Base Programming for SAS 9

- Item 1 : A
- Item 2 : D
- Item 3 : B
- Item 4 : E
- Item 5 : B

SAS Advanced Programming for SAS 9

- Item 1 : D
- Item 2 : D
- Item 3 : B
- Item 4 : A

コラム「SAS四方山話」

第5回 「アンケート四方山話(3) アンケートバイアスと基本集計の方法」

大橋 涉

東京医科歯科大学 医歯学総合研究科大学院 生命情報学
情報医科学センター 臨床・生物統計グループ 特任助教

1)いよいよ集計...の前に、最近見たアンケートの中から調査票の作成からデータの入力方法まで一通り説明させて頂きまして、今度こそSASの出番となります...がその前に、私が最近見たアンケートの中でかなり印象深かったものを2、3ご紹介させて頂きたいと思います。アンケートを見て真っ先に思うことは、多くの方にとっては「どう回答するか」だと思うのですが、そろそろ皆様は、「相手が何を知りたがっているのだろうか?」「このアンケートの目的は何だろうか?」などと考えることもできる頃だと思います。当然、まずい部分の指摘もできる頃ではないでしょうか?というわけで下記の2点について、最低5分は立ち止まって考えてみてください。

Q1: 政府は定額給付金を強行しようとしておりますが、あなたは定額給付金の給付に賛成ですか、反対ですか?

Q2: 小学校における性教育は早すぎないと思いますか?

思う ・ 思わない

まずQ1ですが、「強行」という言葉は既にマイナスイメージのある言葉です。設問に「いかにも定額給付金の施行はいけないことである」というような意味を含んでしまうため、回答そのものが誘導されてしまう可能性があります。この場合、句読点以前の説明は不要...というよりはむしろ付与してはいけません。Q2は表現が紛らわしく、小学校における性教育が早すぎると思っている人でもうっかり「思う」に印を付けてしまう可能性があります。聞かざれば「早すぎると...」ですね。英語の文法で言えば「付加疑問文(Don't you ~?)」や「Do you mind ~ing?(~することを嫌がりませんか)」のような聞き方をすべきではないということです。

実はQ1の誘導については、直接的でなくとも知らずにやってしまうことがあります。例えば顧客の意向が知りたいのであれば、お客様向けアンケートに実施企業名を出すのは何等問題はないのですが、一般の人々を対象とする場合はいかがでしょうか?「正直に回答しにくいからつい模範解答を」「この企業・団体は嫌いだからボロクソに」という心理が働いてしまうのはよくある話であり、実施段階で既にバイアスがかかっています。本当に正確な情報を知りたいのであれば、バイアス防止のためにもアンケートの主権者名を出すべきではないということなのです。某婦人団体がサッカーくじ導入の反対運動をやっている会場で、道行く人に実施したアンケートの結果には、その時点でバイアスがかかっているのは言うまでもありません^{*)}。これほど極端ではないにしても、知らず知らずに回答を誘導してしまう事例はよくある話ですね。特に面接調査やホールテスト、グループインタビューの担当者の皆様は要注意です。

*)このような事例は谷岡一郎先生の著作『「社会調査」のウソ リサーチ・リテラシーのすすめ (文春新書)』に記載されています。統計学云々ではなく、とにかく「読むとスカッとすること」は間違いありません。谷岡先生曰く「統計をいかにわしく用いている人々にとっては、血管が切れるほど頭にくると思うので読まないこと?」をお勧めしているようですが(笑)

2) SASへのデータ読み込み

ようやく本題です。前回までにアンケート結果をMicrosoft Excelにまとめてあるかと思いますが、まずはそれをSASに読み込むことから始めましょう。ソフトウェアを購入して、有料でインストールまではしてもらったけど...データの読み込み方がわからず挫折というパターンは他の統計ソフトウェアでも結構多いので、是非「たかがデータの読み込み」などと考えないで下さい。読み込みには、Enterprise Guide、「ファイル」メニューからインポートウィザード、proc importのうちどれを用いても構いませんが、ここではproc importを用いてみましょう。詳細はSASのトレーニングコース「SASプログラミング」などもございますが、簡単に紹介させていただきます。

```
proc import out=Question(データセット名。任意です)
  datafile=(どこにある何というExcelファイル?)
  sheet="(importしたいワークシート名)"
  dbms=(Excelの何年バージョンですか?);
run;
```

これでwork上にSASデータセット「Question」が完成しているか確認してみましょう^{*)}。完成していたらダブルクリックして変数名を確認してみてください。変数名は「日本語を用いていなければ」Excelで指定したままになっていると思います。

| NUM | Q1 | Q2 | Q3 | Q4 | Q5 | Q6_1 | Q6_2 | Q6_3 | Q6_4 | Q6_5 | Q6_01 | Q7 |
|------|----|----|----|----|----|------|------|------|------|------|-------|------|
| 0001 | 28 | 1 | 0 | 3 | 2 | 1 | 1 | 0 | 1 | 0 | | |
| 0002 | 35 | 0 | 1 | 2 | 3 | 0 | 0 | 0 | 0 | 1 | 教本の多さ | 特になし |

^{*)}完成していない場合はlogウィンドウを確認してみましょう。エラーの原因が示されているはずですが(どのようなプログラムでも、実行したら必ずlogを見ること!...は、筆者も昔先輩に教わりました)

3) 基本的な集計方法

```
Q1 : 年齢 ( ) Q2 : 性別 男・女
Q3 : 過去に統計ソフトウェアの利用経験がある? はい・いいえ
Q4 : 居住地域 1.北海道... (中略) ...8.九州・沖縄
Q5 : この統計ソフトウェアの満足度は?
      1.非常に高い 2.高い方 3.低い方 4.非常に低い
Q6 : 統計ソフトウェアを購入するとき何を重視するか? (該当するもの全て)
      1.価格 2.機能の多さ 3.使いやすさ 4.ユーザー数 5.その他( )
Q7 : 統計ソフトウェア全般に対するご意見
      ( )
```

では、前回のファイルを例に基本集計について説明させていただきます。結論から申しますと、データに合致した集計・解析を行わなければ結果は全く意味をなさないものになります。例えばQ5の平均値などは意味をなしません(カテゴリ数がかつ多ければ傾向ぐらいは見えるかもしれませんが...)ので、この中で平均値を算出して意味があるものはQ1だけです。proc univariate data=Question; var Q1; run; とするだけで、平均値をはじめとした大量の情報が出てまいります。何も指定しなければアンケート全体の年齢が算出されるだけです。例えば男女別であれば「class Q2;」を、地域別であれば「class Q4;」を1行、run;までの間に挿入してください。もしも「男女・地域別の年齢」であれば、by Q2; class Q4;になります。ただし、by~を用いるときには必ずby~の変数にてsortするのを忘れずに(SAS文法全てにおいてこれは共通項です)。また基本統計量だけならば、univariateの

代わりにmeansを用いて行ってもかまいません(実は、procの後を一箇所書き換えるだけでそのまま使えたりするプロシージャがSASには結構存在するのです)。

```
proc univariate data=Question(どのデータを用いる?);
  var Q1;(何について集計したい?);
  class Q2(性別で。もしも地域別ならclass Q4。);
run;
( univariateの代わりにmeansでも構いません )
```

```
proc freq data=Question(どのデータを用いる?);
  tables Q2*Q4;(これは性別*地域);
run;
```

Q2~Q6に関してはfreqプロシージャを用いて「頻度集計」を行います。proc freq data=Question; tables Q2; run; とすれば、男女の人数と割合が集計されます。Q2の代わりにQ3とすればソフトウェアの利用経験別、Q4とすれば地域別に集計されます。もしも、男女・地域別の集計であれば、2行目を「tables Q2*Q4;」と書き換えてください。マーケティングなどではおなじみの、いわゆる「クロス集計」の機能がこれです。結果のみであれば、outputウィンドウでクロス表の形で確認できますが、綺麗なレポートとして出力したいのであればODSなど多彩な機能で対応できます。

アンケートに限らずですが、統計処理において必要なことは「データに合致した処理を行なうこと」に他なりません。それは、次回に説明させていただきます「検定」におきましても全く同様なのです。

Q&A

DDEを使用した上付き文字の出力
MIGRATE プロシジャ実行時のメッセージ
SAS データセットの作成日付をコピーしたい
SAS インストールデータファイルは正しくありません
ログをファイルとウィンドウの両方に出力したい
変数の値を引用符で括りマクロ変数に格納する
Excel へのデータエクスポート時の出力形式
サポートベクターマシンに対応した分析機能
偏プロット(部分プロット)上の値をデータセットに出力
カイ2乗検定における警告メッセージの出力

Q

DDE を使用して EXCEL ファイルに上付き文字 (10 の 2 乗など) を出力する方法はありますか？

A

EXCEL の上付き文字は、通常セルの書式設定にて指定します。このセルの書式設定を行う DDE コマンドは FONT.PROPERTIES に該当し、第 5 引数、第 13 引数と第 14 引数を指定することで上付き文字へ変更することが可能です。各引数の説明は次の通りです。注意点といたしまして、事前にセルの表示形式を文字列に設定する必要があります。

| | |
|---------|--------------------|
| 第 5 引数 | 1 を設定することで上付き文字にする |
| 第 13 引数 | 開始位置 |
| 第 14 引数 | 長さ |

次に、10 の 2 乗を表示するサンプルプログラムをご紹介します。

```

OPTIONS NOXWAIT NOXSYNC;
                                /* EXCEL の起動 */
DATA _NULL_;
  X "START EXCEL";
  X = SLEEP(5);
RUN;

                                /* セルの書式を文字列にする */
FILENAME format DDE 'EXCEL|SYSTEM';
DATA _NULL_;

FILE format;
  PUT '[FORMAT.NUMBER("@")]';
RUN;

                                /* データ入力 */
FILENAME xls DDE "EXCEL|SHEET1!R1C1" NOTAB;
DATA _NULL_;
  FILE xls;
  PUT '102';
RUN;

                                /* 上付き文字設定 */
FILENAME super DDE 'EXCEL|SYSTEM';
DATA _NULL_;
FILE super;
  PUT '[SELECT("R1C1")]';
  PUT '[FONT.PROPERTIES("ARIAL","BOLD",8,0,1,0,0,0,1,0,0,1,3,1)];
RUN;

```

その他の引数の詳細については、Microsoft 社のホームページより MACROFUN.EXE という名前で検索し、EXCEL4 のマクロに関する HELP ファイルにてご確認ください。

Q

SORT プロシジャの NODUPKEY オプションを指定して作成したデータセットを、MIGRATE プロシジャにて移行したところ、以下のメッセージが出力されました。どのように対応すればよいのでしょうか？

NOTE: Migrating INLIB.TEST to OUTLIB.TEST (memtype=DATA).
WARNING: OUTLIB.TEST.DATA の NODUPKEY 並べ替え順序表明は削除されました。再表明するには、NODUPKEY を使い、もう一度並べ替えてください。

A

MIGRATE プロシジャにてデータセットの移行を行なうと、データセット中の NODUPKEY の情報が削除されるため、メッセージが表示されます。

SORT プロシジャの NODUPKEY オプションを使用してデータを再作成するか、MIGRATE プロシジャにて KEEP NODUPKEY オプションを使用して、再度移行作業を実施してください。

```

PROC MIGRATE IN=inlib OUT=outlib KEEP NODUPKEY;
RUN;

```

Q

SAS データセットを COPY プロシジャでコピーしたら、SAS データセットの作成日付が最新の日付になってしまいました。元の作成日付に戻したいのですが、どうしたらよいのでしょうか？

A

SAS 9 より、COPY プロシジャのオプションとして、元のファイルの作成日付を引き継ぐ DATECOPY が利用可能です。

以下の例をご確認ください。

```

                                /* コピー元 */
LIBNAME alib "c:\%a";

                                /* コピー先 */
LIBNAME blib "c:\%b";

                                /* DATECOPY オプションでコピー */
PROC COPY IN=alib OUT=blib DATECOPY;
RUN;

```

詳細については、オンラインヘルプをご参照ください。

<http://support.sas.com/onlinedoc/913/getDoc/ja/proc.hlp/a000085782.htm>

Q

SAS 9 をインストールする際、SAS インストールデータを指定したところ、「SAS 9.1 の SAS インストールデータファイルは正しくありません。」と表示されインストールが行えません。どのように対処すればよいのでしょうか？

A

SAS インストールデータファイルは電子メールにて送付しております。メールシステムによっては、ファイルの改行文字やファイル末尾の空白が欠落し、正しいインストールデータとして認識されない場合があります。

この現象が発生したときは、一度次の操作をお試しください。

1. Windowsのメニュー「スタート|プログラム|アクセサリ|ワードパット」から「ワードパット」を起動します。
2. 「ワードパッド」から、SASインストールデータファイルを開きます。
3. ファイルの末尾に数行の「改行」を追加します。

例

```
...
$_Stream=----- END -----
[ _EOF_ ]
(改行を追加)
(改行を追加)
```

4. SASインストールデータファイルを「上書き保存」し、「ワードパッド」を終了します。
5. 再度、インストールをお試しください。

なお、この現象はライセンス更新の際にも発生する場合があります。その場合も、上記と同様の操作をご確認ください。また、この方法を適用しても依然としてインストールを継続できないときは、テクニカルサポートまでご連絡ください。

Q

特定のステップの実行ログを、ウィンドウとファイルの両方に出力する方法はありますか。

A

一度に両方へ出力する方法はありませんが、PRINTTOプロシジャでファイルに出力した後、そのファイルを読み込んでログウィンドウに出力することで、期待する結果が得られます。

```
/* ログの出力先をファイルにする */
PROC PRINTTO LOG="c:\temp\test.log";
RUN;

/* ログに記録したい処理 */
DATA _NULL_;
RUN;

/* ログの出力先をウィンドウにする */
PROC PRINTTO LOG=LOG;
RUN;

/* ログファイルを読み込んでウィンドウに出力 */
DATA _NULL_;
  INFILE "c:\temp\test.log" LRECL=200 PAD;
  INPUT MSG $CHAR200.;
  PUT MSG;
RUN;
```

Q

ある変数の値をそれぞれ引用符ではさんで、一つのマクロ変数に格納したいのですが、どのようにすればよいでしょうか？

A

QUOTE関数を使用することで値を引用符ではさむことができます。SQLプロシジャのINTO句とこの関数を組み合わせることで実現可能です。

例

```
/* トランザクションデータの作成 */
DATA trans;
  INPUT name $;
DATALINES;
山田
鈴木
;
RUN;

/* トランザクションデータを引用符で括り */
/* スペース区切りでマクロ変数に格納 */
PROC SQL NOPRINT;
  SELECT QUOTE(TRIM(LEFT(name)))
  INTO :list SEPARATED BY ' '
  FROM trans;
QUIT;

/* マスターデータセットの作成 */
DATA master;
  INPUT name $;
DATALINES;
山田
鈴木
高橋
;
RUN;

/* マクロ変数と合致する値を PRINT する */
PROC PRINT DATA=master;
  WHERE name IN (&list);
RUN;
```

Q

EXPORTプロシジャでExcelへ出力した際に、SASデータセットの変数に指定した出力形式が反映されません。出力形式を引き継ぐ方法はありますか？

A

EXPORTプロシジャからのExcel出力では、SASデータセットの変数に指定した出力形式は反映されません。しかし、出力したExcelファイル上で数値にカンマ形式が適用された形で参照するには、以下のような方法があります。

1. タブ区切りのテキストファイルとして出力し、その出力ファイルをExcelで開く。
2. 該当のSASデータセットのビューを作成し、そのビューをExcelに出力する。
3. Excelが提供するDDEの機能を使用して、Excelへの書出しを行なう。
*DDEの使用に関する詳細は、以下のURLにて公開されています。

<http://support.sas.com/onlinedoc/913/getDoc/ja/hostwin.hlp/ddeyntax.htm>
<http://support.sas.com/onlinedoc/913/getDoc/ja/hostwin.hlp/ddeexamples.htm>

```
DATA sample;
  INPUT d1;
  FORMAT d1 COMMA10.2;
  CARDS;
11111.11
22222.22
;
RUN;

/* 1タブ区切りのテキストファイルとして出力 (テキストファイルを Excel で開く) */
PROC EXPORT DATA=sample
  OUTFILE='c:\temp\sample1.txt'
  DEBMS=TAB
  REPLACE;
RUN;

/* 2 ビューを作成し、ビューを Excel ファイルに出力 */
PROC SQL;
```

```

CREATE VIEW work.sample_view AS
SELECT PUT(d1, COMMA10.2) AS d1 FROM sample;
QUIT;

PROC EXPORT DATA=work.sample_view
OUTFILE='c:\temp\sample2.xls'
DBMS=EXCEL
REPLACE;
RUN;

/*3 DDE 出力*/
/* Excel の sheet をオープンして、入力する範囲をマークします。*/
/* Excel のメニューから [ 編集 ] [ コピー ] を選択します。*/
/* SAS のメニューから [ ソリューション ] [ アクセサリ ] [ DDE トリプレット ] を選択します。*/
/* DDE トリプレットが表示されますので、Cnt1 + C キーを押します。*/
/* Cnt1 + V キーを押して、DDE トリプレットを Program Editor に貼り付けます。*/

FILENAME excl DDE 'Excel|[Book1]Sheet1!R1C1:R3C2'; /*!DDE トリプレット指定*/

DATA _NULL_;
SET sample;
FILE excl;
PUT d1 :COMMA10.2;
RUN;

```

Q

サポートベクターマシン(SVM)に対応した分析機能はありますか？

A

SAS Enterprise Miner 6.1からは、2値ターゲットの場合をサポートしたSVMノードが新たに正規版としてサポートされています。

それ以外のプロダクトでは実装されていません。ただし、本質的には2次元最適化に帰着されることから、SAS/ORのプロシジャやSAS/IMLの非線形最適化サブルーチンを利用すれば、問題によってはプログラムを具体的に記述することにより実現が可能です。

Q

REGプロシジャのMODELステートメントにて、PARTIALオプションを指定することによって、偏プロット(部分プロット)が作成されます。このグラフにおける各点の値をデータセットに出力できますか。

A

SAS 9.2では、新たにPARTIALDATAオプションが追加され、偏プロット(部分プロット)における各点の値がアウトプット画面に表示されます。このため、ODS OUTPUTステートメントを用いて、データセットに出力することができます。

SAS 9.1.3もしくは以前のリリースでは、対応しているオプションがありません。しかしながら、偏プロット(部分プロット)はある変数の効果を考慮した上での、2変数間の散布図であることより、以下のようなREGプロシジャのプログラムを用いることによって、値を求めることができます。

例

```

PROC REG DATA=sashelp.class;
MODEL weight height=age;
OUTPUT OUT=out R=r_w r_h;
RUN;
QUIT;

```

この例では、変数AGEを考慮した上での、変数WEIGHT、HEIGHTの偏プロット上の点(R_W, R_H)をデータセットOUTに出力しています。

Q

FREQプロシジャのTABLESステートメントにて、CHISQオプションを指定した際、カイ2乗検定に関して、以下のWARNINGメッセージが表示される場合があります。

WARNING:セルのxx%において、期待度数が5より小さくなっています。カイ2乗検定は妥当な検定でないと思われる。

このメッセージはどのような条件の場合、表示されるのでしょうか。また、このWARNINGメッセージの有無をデータセットに出力できますか。

A

該当のWARNINGメッセージは、期待度数が5より小さいセルの比率が20%より大きな場合に表示されます。設定に応じてアウトプットまたはログ画面に表示されますが、データセットに出力するためのオプションはありません。

該当のWARNINGメッセージの有無をデータセットにて保持するためには、以下のようなプログラムを別途記述することになります。

例

```

/* 期待度数をデータセットに出力 */
PROC FREQ DATA=one NOPRINT;
TABLES row*col /OUTEXPECT SPARSE OUT=out;
WEIGHT count;
RUN;

/* WARNINGの有無を判断 */

DATA flag;
SET out END=eof;
IF EXPECTED < 5 THEN less+1;
IF eof THEN DO;
ratio=less / _N_;
IF ratio > 0.20 THEN flag=1; ELSE flag=0;
OUTPUT;
END;
KEEP ratio flag;
RUN;

```

SAS Training

SASトレーニングのお知らせ

特別トレーニングコースの開催のご案内

「医学研究における経時データ解析」コース(2日間)

日 程：2009年5月25日(月)～26日(火) 10:00～17:00(東京会場)

価 格：147,000円(税込)/チケットのお取り扱いはありません。

受講対象：SASによる統計解析に従事する方

前提知識：「医薬向け分散分析」を受講済みか、同程度の知識のある方

学習内容：医薬統計専門家により以下の内容で行います。

多くの医学研究においては、研究対象者ごとに時間を追って反応変数を繰り返し測定することがあり、このような研究は経時観察研究などと呼ばれます。

その研究目的の一つは、反応変数と時間、治療効果あるいは他の説明変数との関係を定量的に評価することです。同一対象者内の反応変数には相関が生じることから、統計的に妥当な推論を行うためには測定単位に起因する相関を考慮する必要があります。そのような反応変数間の相関を考慮する方法として、線型混合モデル及び一般化推定方程式(GEE)による解析方法が近年注目を浴びてきています。このコースでは、線型混合モデルと一般化推定方程式の理論的背景について講義し、実際の医学データの解析を通してMIXEDプロシジャ及びGENMODプロシジャの使用法について学びます。

- ・線型混合モデル及び一般化推定方程式(GEE)に関する基本概念の説明
- ・MIXEDプロシジャ及びGENMODプロシジャの使用法
- ・医学分野における実データの解析

コンテンツ： 経時データ解析へのイントロダクション

- ・経時データ解析の基本概念
- ・探索的データ解析
- 連続型反応変数に対する経時データ解析
- ・線型混合モデル
- ・共分散構造の評価
- ・モデルの構築と解釈
- ・ランダム係数モデル
- ・モデルの評価
- 離散型反応変数に対する経時データ解析
- ・二値反応変数に対するGEE回帰モデル
- ・順序反応変数に対するGEE回帰モデル
- ・GEE回帰モデルの回帰診断
- ・一般化線型混合モデル

新規トレーニングコースの開催のご案内

SAS® Enterprise Guide®による回帰分析」コース(2日間)

日 程：2009年6月29日(月)～30日(火) 10:00～17:00(東京会場)

価 格：115,500円(税込)/チケット捺印数2

受講対象：ポイント&クリックで操作するSAS Enterprise Guideを使用して、統計解析の代表的手法である回帰分析を行いたい方

前提知識：「SASによる統計解析」または「SAS Enterprise Guideによるデータ解析入門」を受講済みか、同程度の知識のある方

学習内容：回帰分析、およびロジスティック回帰についての基本的な考え方と手法を紹介し、コース内容は、「SASによる回帰分析」と同等です。

コンテンツ： 回帰分析

- ・探索的データ解析
- ・単回帰分析
- ・重回帰分析の考え方
- ・ステップワイズ回帰のテクニック
- モデルの構築と解釈
- ・総当たり法による変数選択を伴う回帰
- ・残差の分析
- ・影響力の大きいオブザベーション
- ・多重共線性
- ロジスティック回帰によるカテゴリカルデータ解析
- ・ロジスティック回帰入門
- ・多重ロジスティック回帰
- ・ロジットのプロット(自己学習)

「SASプログラミング1：必須要素」コース（3日間）

日程：2009年6月3日(水)～5日(金) 10:00～17:00(東京会場)

価格：173,250円(税込)/チケット捺印数3

本コースの内容はSAS 9.2をベースとしていますが、それ以前のバージョンを使用しているユーザーにも適しています。

概要：本コースは、SASプログラムの記述方法を学習したいユーザーのためのコースです。SASプログラミングを学習するための入り口となり、他の多くのSASコースへの前提条件となります。SASプログラムを記述する必要がなく、ポイント&クリックインターフェイスを使用したい場合には、「SAS Enterprise Guideによるデータ加工とレポート作成」コースの受講をお勧めします。

本コースは、SAS認定試験「SAS Certified Base Programmer Credential for SAS 9」の準備にも適しています。

受講対象：SASプログラムの記述を始めようとしているすべてのユーザー

前提知識：本コースの受講には、コンピュータソフトウェアの使用経験が必要です。特に次のことが必要になります。

- ・お使いのオペレーティングシステム上のファイル構造やシステムコマンドを理解していること
- ・お使いのオペレーティングシステム上のデータファイルにアクセスすること

SASの経験は必要ありません。もし、この前提条件を満たしていないか、プログラミングの経験が浅く、コース内容が多いと感じられる場合は、このコースの受講の前に「SASプログラミング入門：基本概念」コースを受講することをお勧めします。「SASプログラミング入門：基本概念」コースでは、コンピュータプログラミングの初心者のために「SASプログラミング1：必須要素」コース内容の一部を丁寧に解説しています。

- 学習内容：
- ・SASウィンドウ環境を紹介します。
 - ・SASデータセットにさまざまな種類のデータを読み込みます。
 - ・SASデータセットを検証し、クリーニングします。
 - ・SAS変数やデータのサブセットを作成します。
 - ・SASデータセットを結合します。
 - ・リストレポートや要約レポートの作成や拡張を行います。

コンテンツ：イントロダクション

- ・コース概要
- ・Foundation SASの概要
- SASを始めよう
- ・SASプログラミング イントロダクション
- ・SASプログラムの実行
- SAS構文での作業
- ・基本概念のマスター
- ・構文エラーの診断と修正
- SASデータセットに親しもう
- ・ディスクリプタ部とデータ部の調査
- ・SASデータライブラリへのアクセス
- ・リレーショナルデータベースへのアクセス(自己学習)
- SASデータセットの読み込み
- ・データ読み込み イントロダクション
- ・入力としてのSASデータセットの利用
- ・オブザベーションと変数のサブセット化
- ・永久属性の追加
- Excelワークシートの読み込み
- ・入力としてのExcelデータの利用
- ・Excelワークシートですらに行うこと(自己学習)

区切り記号付きローデータファイルの読み込み

- ・入力としての標準区切り記号付きデータの利用
- ・入力としての非標準区切り記号付きデータの利用
- データの検証とクリーニング
- ・データの検証とクリーニング イントロダクション
- ・ローデータファイルを読み込んだ場合のデータエラーの調査
- ・PRINTプロシジャとFREQプロシジャによるデータの検証
- ・MEANSプロシジャとUNIVARIATEプロシジャによるデータの検証
- ・不正データのクリーニング
- データ操作
- ・変数の作成
- ・条件付き変数の作成
- ・オブザベーションのサブセット化
- SASデータセットの組み合わせ
- ・データセットの組み合わせ イントロダクション
- ・データセットの追加
- ・データセットの縦結合
- ・1対1のデータセットのマージ
- ・1対多のデータセットのマージ
- ・一致しない行のあるデータセットのマージ
- レポートの拡張
- ・グローバルステートメントの利用
- ・ラベルと出力形式の追加
- ・ユーザー定義出力形式の作成
- ・オブザベーションのサブセット化とグループ化
- ・外部ファイルへの直接出力
- 要約レポートの生成
- ・FREQプロシジャの利用
- ・MEANSプロシジャの利用
- ・TABULATEプロシジャの利用(自己学習)
- SAS/GRAPH[®]を利用したグラフ イントロダクション
- ・棒グラフと円グラフの作成
- ・散布図の作成
- ・出力の拡張

SAS Institute Japan株式会社では、今後も多岐にわたったトレーニングコースを追加していく予定です。コース内容・日程等の詳細は、順次弊社Webサイトに公開しますので、以下のURLをご参照ください。
<http://www.sas.com/japan/training/>

その他、トレーニングに関する情報については、上記のURLをご参照いただくか、下記トレーニング担当までお問い合わせください。

トレーニングに関するお問い合わせ先

- TEL 03-3533-3835
- FAX 03-3533-3781
- E-mail JPNTraining@sas.com

Latest Releases

最新リリース情報

PCプラットフォーム

Windows版 SAS 9.1.3 / 9.2
64-bit Windows(Itanium)版 SAS 9.1.3 / 9.2

UNIXプラットフォーム

SunOS/Solaris版 SAS 9.1.3 / 9.2
HP-UX版 SAS 9.1.3 / 9.2
HP-UX(Itanium)版 SAS 9.1.3 / 9.2
AIX版 SAS 9.1.3 / 9.2
Linux(Intel)版 SAS 9.1.3 / 9.2

メインフレームプラットフォーム

IBM版(OS/390、z/OS) SAS 9.1.3 / 9.2
富士通版(F4、MSP) SAS 6.09E TS470
日立版(VOS3) SAS 6.09E TS470

9.2はIntelligence Platform製品を除く

Information

SAS Technical News 送付についてのご案内

SAS Technical Newsは次の方を対象にお送りしています。

- ・ SASコンサルタントとしてご登録の方
- ・ SAS Technical Newsの購読をお申し込みいただいている方

今後SAS Technical News購読が不要の方、配信先の変更等をご希望の方は、下記URLよりお手続きください。

配信停止

「定期購読キャンセルのお申し込み」よりお手続きください。

<http://www.sas.com/japan/corporate/material.html#tn>

配信先変更手続き

「お問い合わせ・ご依頼受付ページ」よりお手続きください。

http://www.sas.com/japan/sasj_privacy.html#inquiry

SAS Technical News Spring 2009

発行
SAS Institute Japan株式会社

テクニカルニュースに関するお問い合わせ先

テクニカルサポートグループ
TEL:03-3533-3877
FAX:03-3533-3781



SAS Institute Japan株式会社 www.sas.com/japan/

| | |
|--------------------|--------------------|
| 東京本社 | 大阪支店 |
| 〒104-0054 | 〒530-0004 |
| 東京都中央区勝どき1-13-1 | 大阪市北区堂島浜1-4-16 |
| イヌイビル・カチドキ | アクア堂島西館 12F |
| Tel 03 (3533) 6921 | Tel 06 (6345) 5700 |
| Fax 03 (3533) 6927 | Fax 06 (6345) 5655 |